21 Chapter: Algebraic Extensions

In this chapter we discuss how to create algebraic extensions in GAP. The polynomials $x^5 - 7$ is irreducible over $\mathbf{Q}$:

```
gap> x:= X(Rationals, "x");
x
gap> f:= x^5-7;
x^5-7
gap> Factors(x^5-7);
[ x^5-7 ]
```

Thus if we adjoin a zero of this polynomial to $\mathbf{Q}$ we get a field of degree five over $\mathbf{Q}$.

```
gap> F:=AlgebraicExtension(Rationals,f);
<algebraic extension over the Rationals of degree 5>
gap> a:= RootOfDefiningPolynomial(F);
(a)
```

The first command above defines a field F that is obtained by adjoining a zero of $x^5 - 7$ to $\mathbf{Q}$. The second command assigns the name a to a zero of f. (Thus $F = \mathbf{Q}(a)$.) Every element in $\mathbf{Q}(a)$ can be written in the form $q_0 + q_1 a + q_2 a^2 + q_3 a^3 + q_4 a^4$ for $q_i \in \mathbf{Q}$. We can now find the minimal polynomial of linear combinations of a over $\mathbf{Q}$. For example, the following finds the minimal polynomial of $4(7^{1/5}) + 10$ over $\mathbf{Q}$.

```
gap> MinimalPolynomial(Rationals, 4*a+10);
x^5-50*x^4+1000*x^3-10000*x^2+50000*x-107168
```

Exercises

21.1 Use GAP to find the minimal polynomial of $\sqrt[3]{2}+\sqrt[3]{4}$ over $\mathbf{Q}$. [Gallian, Chapter 21, Exercise 16]

21.2 Use GAP to find the minimal polynomial of $5 + 4(\sqrt[3]{2}) + 10(\sqrt[3]{4})$ over $\mathbf{Q}$.

21.3 By hand find the minimal polynomial of $1 + i$ over $\mathbf{Q}$. Check your work using GAP.

We can also set up a finite field of order p^n by adjoining a root of an irreducible polynomial of degree n over $GF(p)$ to $GF(p)$. For example, the following creates the field of order 27 by adjoining a root of an irreducible cubic polynomial over $GF(3)$ to $GF(3)$:

```
gap> r:= GF(3);;
gap> x:= X(GF(3),"x");;
gap> f:= x^3 + 2*x^2 + 1;
x^3-x^2+Z(3)^0
gap> IsIrreducible(f);
true
gap> F:= AlgebraicExtension(r, f);
<field of size 27>
```


We can then use GAP to convert from multiplicative to additive notation in this field. [See Gallian, Chapter 22, Table 22.1]

```
gap> a:= RootOfDefiningPolynomial(F);
(a)
gap> a^3;
(Z(3)+a^2)
gap> a^4;
(Z(3)+Z(3)*a+a^2)
gap> a^6+Z(3)^0;
(Z(3)^0+Z(3)*a+Z(3)*a^2)
```

Careful: Recall GAP denotes the number 1 in this field by $Z(3)^0$ and the number 2 by $Z(3)$.

